Solve Constained minimization using KKT

```
> restart:
```

Minimize

```
> f := x*y ;
```

$$f := x\,y \quad (1)$$

Constraints

```
> g1 := x^2+y^2 <= 1 ;
```

$$g1 := x^2 + y^2 \leq 1 \quad (2)$$

```
> g2 := x^2 <= y ;
```

$$g2 := x^2 \leq y \quad (3)$$

Rewrite constraints ad gk(x,y)>=0

```
> g1 := rhs(g1) - lhs(g1) ;
```

$$g1 := -x^2 - y^2 + 1 \quad (4)$$

Add variable `b` for constraint g2

```
> g2 := rhs(g2) - lhs(g2) ;
```

$$g2 := -x^2 + y \quad (5)$$

Step 1, build Lagrangian and non linear system

```
> L := f-mu1*g1-mu2*g2 ;
```

$$L := -\left(-x^2 - y^2 + 1\right)\mu 1 - \left(-x^2 + y\right)\mu 2 + x\,y \quad (1.1)$$

Nonlinear system is the gradient of the Lagrangian

```
> EQ1 := diff(L,x) ;
  EQ2 := diff(L,y) ;
  EQ3 := mu1*g1 ;
  EQ4 := mu2*g2 ;
```

$$EQ1 := 2\,\mu 1\,x + 2\,\mu 2\,x + y$$
$$EQ2 := 2\,\mu 1\,y - \mu 2 + x$$
$$EQ3 := \left(-x^2 - y^2 + 1\right)\mu 1$$
$$EQ4 := \left(-x^2 + y\right)\mu 2 \quad (1.2)$$

Step 2: solve the nonlinear system

Case mu1=mu2=0, i.e. we are inside of the contraints

```
> EQRED1 := subs( mu1=0, mu2=0, [EQ||(1..4)]): <%> ;
```

$$\begin{bmatrix} y \\ x \\ 0 \\ 0 \end{bmatrix} \quad (2.1)$$

Solution 1:

```
> x_sol1, y_sol1, mu1_sol1, mu2_sol1 := 0, 0, 0, 0 ;
```
$$x\_sol1, y\_sol1, mu1\_sol1, mu2\_sol1 := 0, 0, 0, 0 \quad \textbf{(2.2)}$$

Case mu1<>0, mu2<>0, i.e. we are ON the contraints

```
> EQRED2 := subs( mu1=1, mu2=1, [EQ||(3..4)]): <%> ;
```
$$\begin{bmatrix} -x^2 - y^2 + 1 \\ -x^2 + y \end{bmatrix} \quad \textbf{(2.3)}$$

```
> EQRED2[1]-EQRED2[2] ;
  YSOL := solve( %, {y} ) ;
```
$$-y^2 - y + 1$$
$$YSOL := \left\{y = -\frac{1}{2}\sqrt{5} - \frac{1}{2}\right\}, \left\{y = \frac{1}{2}\sqrt{5} - \frac{1}{2}\right\} \quad \textbf{(2.4)}$$

Discard y negative for real solution

```
> y_sol2 := subs(YSOL[2],y) ;
  x_sol2 := sqrt(y_sol2) ;
```
$$y\_sol2 := \frac{1}{2}\sqrt{5} - \frac{1}{2}$$
$$x\_sol2 := \frac{1}{2}\sqrt{-2 + 2\sqrt{5}} \quad \textbf{(2.5)}$$

```
> subs( x=x_sol2, y=y_sol2, {EQ||(1..2)} ) ;
  solve( %, {mu1, mu2} ) ; evalf(%) ;
```
$$\left\{2\left(\frac{1}{2}\sqrt{5} - \frac{1}{2}\right)\mu 1 - \mu 2 + \frac{1}{2}\sqrt{-2 + 2\sqrt{5}}, \sqrt{-2 + 2\sqrt{5}}\,\mu 1 + \sqrt{-2 + 2\sqrt{5}}\,\mu 2 + \frac{1}{2}\sqrt{5} - \frac{1}{2}\right\}$$
$$\left\{\mu 1 = -\frac{3}{10}\frac{\left(\sqrt{5} - 1\right)\sqrt{5}}{\sqrt{-2 + 2\sqrt{5}}}, \mu 2 = -\frac{1}{10}\frac{\left(8 - 4\sqrt{5}\right)\sqrt{5}}{\sqrt{-2 + 2\sqrt{5}}}\right\}$$
$$\{\mu 1 = -0.5273663763, \mu 2 = 0.1342906872\} \quad \textbf{(2.6)}$$

Discard the solution, mu1 < 0

```
> y_sol2 := subs(YSOL[2],y) ;
  x_sol2 := -sqrt(y_sol2) ;
```
$$y\_sol2 := \frac{1}{2}\sqrt{5} - \frac{1}{2}$$
$$x\_sol2 := -\frac{1}{2}\sqrt{-2 + 2\sqrt{5}} \quad \textbf{(2.7)}$$

```
> subs( x=x_sol2, y=y_sol2, {EQ||(1..2)} ) ;
  solve( %, {mu1, mu2} ) ; evalf(%) ;
```
$$\left\{2\left(\frac{1}{2}\sqrt{5} - \frac{1}{2}\right)\mu 1 - \mu 2 - \frac{1}{2}\sqrt{-2 + 2\sqrt{5}}, -\sqrt{-2 + 2\sqrt{5}}\,\mu 1 - \sqrt{-2 + 2\sqrt{5}}\,\mu 2 + \frac{1}{2}\sqrt{5} - \frac{1}{2}\right\}$$

$$\left\{\mu 1 = \frac{3}{10}\,\frac{\left(\sqrt{5}-1\right)\sqrt{5}}{\sqrt{-2+2\sqrt{5}}},\ \mu 2 = \frac{1}{10}\,\frac{\left(8-4\sqrt{5}\right)\sqrt{5}}{\sqrt{-2+2\sqrt{5}}}\right\}$$

$$\{\mu 1 = 0.5273663763,\ \mu 2 = -0.1342906872\} \quad (2.8)$$

Discard the solution, mu1 < 0

Case mu1=0 and second contraint actives

```
> EQRED3 := subs( mu1=0, [EQ||(1..4)]): <%> ;
```

$$\begin{bmatrix} 2\,\mu 2\,x + y \\ -\mu 2 + x \\ 0 \\ \left(-x^2 + y\right)\mu 2 \end{bmatrix} \quad (2.9)$$

```
> mu1_sol2 := 0 ;
```

$$mu1\_sol2 := 0 \quad (2.10)$$

```
> EQQ1 := subs(mu2=1,EQRED3[4]) ;
```

$$EQQ1 := -x^2 + y \quad (2.11)$$

```
> EQQ2 := op(solve ( EQRED3[2], {mu2} )) ;
```

$$EQQ2 := \mu 2 = x \quad (2.12)$$

```
> EQQ3 := subs( EQQ2, EQRED3[1] ) ;
```

$$EQQ3 := 2\,x^2 + y \quad (2.13)$$

```
> solve( {EQQ1,EQQ3}, {x,y} ) ;
```

$$\{x = 0,\ y = 0\} \quad (2.14)$$

Case mu2=0 and first contraint actives

```
> EQRED3 := subs( mu2=0, [EQ||(1..4)]): <%> ;
```

$$\begin{bmatrix} 2\,\mu 1\,x + y \\ 2\,\mu 1\,y + x \\ \left(-x^2 - y^2 + 1\right)\mu 1 \\ 0 \end{bmatrix} \quad (2.15)$$

```
> EQQ1 := subs(mu1=1,EQRED3[3]) ;
```

$$EQQ1 := -x^2 - y^2 + 1 \quad (2.16)$$

```
> EQQ2 := simplify(EQRED3[1]*y-EQRED3[2]*x) ;
```

$$EQQ2 := -x^2 + y^2 \quad (2.17)$$

```
> MU1 := solve(EQRED3[1],{mu1} ) ;
```

$$MU1 := \left\{\mu 1 = -\frac{1}{2}\,\frac{y}{x}\right\} \quad (2.18)$$

```
> solve( {EQQ1,EQQ2}, {x,y} ) ; SOLALL := allvalues({%}) ;
```

$$\{x = RootOf(2\_Z^2 - 1),\ y = RootOf(2\_Z^2 - 1)\},\ \{x = -RootOf(2\_Z^2 - 1),\ y = RootOf(2\_Z^2 - 1)\}$$

$$(2.19)$$

$$SOLALL := \left\{\left\{x = -\frac{1}{2}\sqrt{2}, y = \frac{1}{2}\sqrt{2}\right\}, \left\{x = \frac{1}{2}\sqrt{2}, y = \frac{1}{2}\sqrt{2}\right\}\right\}, \left\{\left\{x = -\frac{1}{2}\sqrt{2}, y = -\frac{1}{2}\sqrt{2}\right\}, \left\{x = \frac{1}{2}\sqrt{2}, y = -\frac{1}{2}\sqrt{2}\right\}\right\} \quad (2.19)$$

```
> subs( SOLALL[1][1], [g1,g2]) : evalf(%) ;
  subs( SOLALL[1][2], [g1,g2]) : evalf(%) ;
  subs( SOLALL[2][1], [g1,g2]) : evalf(%) ;
  subs( SOLALL[2][2], [g1,g2]) : evalf(%) ;
```

$$[0., 0.2071067810]$$
$$[0., 0.2071067810]$$
$$[0., -1.207106781]$$
$$[0., -1.207106781] \quad (2.20)$$

```
> x_sol2   := subs(SOLALL[1][1],x) ;
  y_sol2   := subs(SOLALL[1][1],y) ;
  mu1_sol2 := subs( SOLALL[1][1],subs(MU1,mu1) ) ;
  mu2_sol2 := 0 ;
```

$$x\_sol2 := -\frac{1}{2}\sqrt{2}$$
$$y\_sol2 := \frac{1}{2}\sqrt{2}$$
$$mu1\_sol2 := \frac{1}{2}$$
$$mu2\_sol2 := 0 \quad (2.21)$$

```
> x_sol3   := subs(SOLALL[1][2],x) ;
  y_sol3   := subs(SOLALL[1][2],y) ;
  mu1_sol3 := subs( SOLALL[1][2],subs(MU1,mu1) ) ;
  mu2_sol3 := 0 ;
```

$$x\_sol3 := \frac{1}{2}\sqrt{2}$$
$$y\_sol3 := \frac{1}{2}\sqrt{2}$$
$$mu1\_sol3 := -\frac{1}{2}$$
$$mu2\_sol3 := 0 \quad (2.22)$$

Discard caurse mu1 < 0

Check first solution for second order condition

Compute the gradient of the constraints

```
> gradG1 := <diff(g1,x)|diff(g1,y)>;
  gradG2 := <diff(g2,x)|diff(g2,y)>;
```

$$gradG1 := \begin{bmatrix} -2x & -2y \end{bmatrix}$$
$$gradG2 := \begin{bmatrix} -2x & 1 \end{bmatrix} \quad (3.1)$$

```
> SOL := x=x_sol1, y=y_sol1, mu1 = mu1_sol1, mu2 = mu2_sol1 ;
```

(3.2)

$$SOL := x = 0, y = 0, \mu 1 = 0, \mu 2 = 0 \quad (3.2)$$

Seacrh active constraints

```
> subs( SOL, g1 ) ; # g1 is NOT Active
  subs( SOL, g2 ) ;
```

$$1$$

$$0 \quad (3.3)$$

```
> G  := gradG2 ;
  G1 := subs( SOL, G ) ;
```

$$G := \begin{bmatrix} -2x & 1 \end{bmatrix}$$

$$G1 := \begin{bmatrix} 0 & 1 \end{bmatrix} \quad (3.4)$$

```
> NS := op(LinearAlgebra[NullSpace](G1)) ;
```

$$NS := \begin{bmatrix} 1 \\ 0 \end{bmatrix} \quad (3.5)$$

```
> Z := alpha*NS;
```

$$Z := \begin{bmatrix} \alpha \\ 0 \end{bmatrix} \quad (3.6)$$

Compute the Hessian of the Lagrangian

```
> HessL := <<diff(L,x,x),
              diff(L,x,y)>|
             <diff(L,y,x),
              diff(L,y,y)>>;
```

$$HessL := \begin{bmatrix} 2\mu 1 + 2\mu 2 & 1 \\ 1 & 2\mu 1 \end{bmatrix} \quad (3.7)$$

```
> HessL1 := subs( SOL,HessL ) ;
```

$$HessL1 := \begin{bmatrix} 0 & 1 \\ 1 & 0 \end{bmatrix} \quad (3.8)$$

```
> LinearAlgebra[Transpose](Z).HessL1.Z ;
```

$$0 \quad (3.9)$$

Check second solution for second order condition

```
> SOL := x=x_sol2, y=y_sol2, mu1 = mu1_sol2, mu2 = mu2_sol2 ;
```

$$SOL := x = -\frac{1}{2}\sqrt{2}, y = \frac{1}{2}\sqrt{2}, \mu 1 = \frac{1}{2}, \mu 2 = 0 \quad (4.1)$$

Seacrh active constraints

```
> subs( SOL, g1 ) ;
  subs( SOL, g2 ) ; # g2 is NOT Active
```

$$0$$

$$-\frac{1}{2} + \frac{1}{2}\sqrt{2} \quad (4.2)$$

```
> G  := gradG1 ;
```

```
G2 := subs( SOL, G ) ;
```

$$G := \begin{bmatrix} -2\,x & -2\,y \end{bmatrix}$$

$$G2 := \begin{bmatrix} \sqrt{2} & -\sqrt{2} \end{bmatrix} \quad (4.3)$$

```
> NS := op(LinearAlgebra[NullSpace](G2)) ;
```

$$NS := \begin{bmatrix} 1 \\ 1 \end{bmatrix} \quad (4.4)$$

```
> Z := alpha*NS;
```

$$Z := \begin{bmatrix} \alpha \\ \alpha \end{bmatrix} \quad (4.5)$$

Compute the Hessian of the Lagrangian

```
> HessL2 := subs( SOL,HessL ) ;
```

$$HessL2 := \begin{bmatrix} 1 & 1 \\ 1 & 1 \end{bmatrix} \quad (4.6)$$

```
> LinearAlgebra[Transpose](Z).HessL1.Z ;
```

$$2\,\alpha^2 \quad (4.7)$$